# 配置
*** Setting ***
Documentation   测试登录页面
Library         SeleniumLibrary
Suite Setup     打开Chrome浏览器到首页
Suite Teardown  关闭浏览器
Test Teardown   截取当前图片

#设置变量 变量名使用 ${} 形式
*** Variables ***
${mainpage}         http://49.233.108.117:3000/
${loginurl}         http://49.233.108.117:3000/signin
${chromedriver}     ./drivers/chromedriver.exe

# 创建测试用例
*** Test Cases ***
用户注册功能验证
    跳转到注册页面
    用户注册    xiaowang123     123456      123456     xiaowang@123.com

用户登录功能测试
    跳转到登录页面
    用户登录    xiaowang123     123456


*** Keywords ***
打开Chrome浏览器到首页
    Open Browser    url=${mainpage}     browser=chrome      executable_path=${chromedriver}
    Set Screenshot Directory    ./screenshots

关闭浏览器
    Close Browser

跳转到注册页面
    Click Element   link:注册

跳转到登录页面
    Click Element   link:登录

用户登录
    [Arguments]     ${username}     ${passwd}
    Input Text      id:name             ${username}
    Input Password  id:pass             ${passwd}
    Click Element   css:[value="登录"]

用户注册
    [Arguments]     ${username}     ${passwd}   ${repasswd}     ${email}
    Input Text  id:loginname    ${username}
    Input Text  id:pass         ${passwd}
    Input Text  id:re_pass       ${repasswd}
    Input Text  id:email        ${email}
    Click Element   css:[value="注册"]

截取当前图片
    Capture Page Screenshot